DAM cursus Klinische Video/Foto Analyse
Opdracht Zweefkip in Excel.

In deze handleiding maakt u kennis met een groot aantal data-bewerkingen in Excel. Het doel van deze training is om u staat te stellen uw eigen video's te analyseren, indien u bewegingsvormen kiest die niet in de Excel-sjablonen van deze cursus zijn behandeld. Om didactische redenen worden de diverse stappen in de berekeningen na elkaar uitgevoerd en niet, zoals in de sjablonen, in één formule gecombineerd. De bedoeling is om 'from scratch' te starten en zelf de markers in het voorbeeldfilmpje te tracken. Mocht u sneller de stap naar deze handleiding willen zetten omdat u vaardig genoeg bent in het tracken van de markers, dan is er via de volgende link (te openen met ctrl+click) ook een voorbeeld databestand beschikbaar dat u kunt gebruiken. In dat geval begint u op pagina 2 van deze handleiding.

Kinovea
Laad de video zweefkip.mkv. Plaats de oorsprong van het coördinatenstelsel in de linker onderhoek van het video-frame.
Teken een lijn van de heupmarker naar de enkelmarker in frame 1 van de video. De afstand tussen de markers bedraagt 90 cm.

Track in de film van de turner de marker op de voet,heup,schouder en hand (in die volgorde).
Exporteer de data in xml formaat.

Kinovea Converter
Open de datafile (xml) in de converter en filter de data met een 5 pt moving average (2 x)

Copy en paste de data (inclusief de kolomheaders) naar een lege Excel file. Zorg dat de data start in rij 3 van de excelfile. Wijzig de naam van het eerste tabblad van de excelfile in: data. Noem blad 2: model. Sla deze file direct even op.

	A	B	C	D	E	F	G	H
1	ankle		hip		shoulder		hand	
2	X	Y	X	Y	X	Y	X	Y
3	145	49	118,99	227,87	176,73	314,39	305	339
4	145	49,09	117,96	227,56	176,82	313,4	305	339
5	145,09	49,35	116,43	226,88	176,93	311,83	305	339
6	145,35	49,68	115,21	225,89	176,98	310,27	305	339
7	145,77	49,9	114,46	224,56	177,09	308,66	305	339



Heup en schouderhoek bepalen

Bij het bepalen van de heup- en schouder hoek in elk frame maken we gebruik van het inproduct (zie presentatie videotechniek van cursusavond 2 van de damcursus klinische video/foto analyse via deze internet link).(open de link met ctrl +re muis click)

$$\alpha = \text{acos}\ \frac{(ax\ .\ bx) + (ay\ .\ by)}{a\ .\ b}$$

formule 1 (gebruik van het inproduct bij het bepalen van de hoek tussen twee vectoren)

Heuphoek

De twee vectoren waartussen we de hoek willen bepalen, zijn de lijnen van heup naar schouder (vector a groen) en van heup naar enkel (vector b rood).



Bij het bepalen van ax en ay geldt de regel: eindpunt- beginpunt. Eerst bepalen we de lengte ax door de x coördinaat van de schouder af te trekken van de x coördinaat van de heup.

Type in cel i2 de titel van deze kolom: ax

In cel i3 noteren we =E3-C3 De lengte ax wordt nu berekend.

	I
	ax
39	57,74
39	58,86

Selecteer de cel en zet de muis op de rechter onderpunt van de cel. Houdt de rechtermuisknop ingedrukt en sleep met de muis tot de laatste rij van kolom i. De berekening wordt nu op alle rijen toegepast.

Tip: gebruik de twee eerste cellen van elke kolom om een omschrijving van de inhoud van de kolom te vermelden

In kolom j bepalen we op identieke wijze de lengte van ay (yschouder - yheup)
In kolom k berekenen we bx (xenkel-xheup) en in kolom L berekenen we by (yenkel-yheup)

De lengte van de twee vectoren hebben we ook nodig om uiteindelijk het inproduct te berekenen. De lengte van vector a komt in kolom M. zet in cel M2 lengte a en plaats in cel M3 de bekende pythagoras formule:
=WORTEL(I3^2+J3^2) voor een Engelstalige versie van Excel =SQRT(I3^2+J3^2)
Bereken zelf de lengte van vector b in kolom N. (Sleep de formule steeds weer naar alle andere cellen in de kolom)

We hebben nu alle ingrediënten om de formule 1 te vullen. De code in cel O3 is als volgt

=BOOGCOS(((I3*K3)+(J3*L3))/(M3*N3)) Let op de haakjes!!! Of (engels)

=ACOS(((I3*K3)+(J3*L3))/(M3*N3)**) Ga na of je formule 1(het inproduct) herkent in deze code**

De hoek staat nu nog in radialen. Omvormen naar graden doen we in kolom P

 In P3 noteren we =GRADEN(O3) of (engels) =DEGREES(O3)

Grafiek van de heuphoek maken

Selecteer de eerste datacel van kolom P (P3) en druk tegelijk op ctrl+shift+end Alle rijen van de kolom P worden nu geslecteerd. Een andere manier om alle gevulde cellen van een bepaalde kolom te selecteren is de volgende: Selecteer een cel en ga met de muis op de rechter onderhoek van de cel staan. De cursor wordt nu een zwart kruisje. Dubbelklik vervolgens op deze positie.



Bekijk een [instructievideo]() in uw webbrowser (link openen met ctrl+click) waarin wordt getoond hoe u de grafiek in Excel kunt maken.
Of lees onderstaande instructie.

Kies in de menubalk van Excel voor invoegen (insert) en vervolgens in de groep grafieken (charts) voor lijn (Line). Kies de eerste optie. Er wordt nu een grafiek van de heuphoek getoond





Om de resultaten en de berekeningen wat gescheiden te houden, willen we de grafiek graag tonen op blad 2 (model). Klik met de rechtermuis op de grafiek en kies voor: grafiek verplaatsen (move chart) en kies in het venster dat nu verschijnt voor object in: model Op de verticale as van de grafiek staat de hoek, op de horizontale as de framenummers. Bekijk in kinovea uw film en bestudeer daarbij de grafiek in Excel.

Schouderhoek



Het bepalen van de schouderhoek is vergelijkbaar met het bepalen van de heuphoek We gaan de hoek bepalen tussen vector a (romp groen) en c (arm blauw). ax en ay hebben we al eerder berekend. Maar de vector a wijst nu de andere kant op. De afstand ax moeten we nu dus uitrekenen als (eindpunt-beginpunt =heupx-schouderx) terwijl we dat in kolom i omgekeerd hebben gedaan. Het omdraaien van de bewerking is eenvoudigweg te realiseren door een – teken voor de uitkomst van kolom i te plaatsen. Dit doen we in kolom Q door in cel Q3 in te voeren: =-i3

In kolom R zetten we =-J3
cx en cy komen in kolom S en T

Zet onderstaande rekenregels zelf om naar Excelcode (raadpleeg zo nodig de appendix)

Voor kolom S: cx =(handx-schouderx)
Voor kolom T cy =(handy-schoudery)

In kolom U berekenen we de lengte van c (arm) *(de lengte van a hebben we al berekend (staat in kolom M). Door het kwadrateren van de waarden is de lengte van a natuurlijk niet gevoelig voor de volgorde waarin we ax en ay bepaald hebben*).

Formule voor de lengte van c (plaats in U3):

=WORTEL(S3^2+T3^2) of engels =SQRT(S3^2+T3^2)

Vervolgens in kolom V de hoek (in radialen) bepalen

=BOOGCOS(((Q3*S3)+(R3*T3))/(M3*U3))

Of in de engelse versie =ACOS(((Q3*S3)+(R3*T3))/(M3*U3))

(tip: kopieer de formule uit cel O3 en pas alleen de verwijzingen naar de cellen aan)

Tenslotte nog omrekenen naar graden (kolom W)

=GRADEN(V3) of (engels) =DEGREES(V3)

Grafiek van de schouderhoek maken
We willen de grafiek van de schouderhoek graag in dezelfde figuur opnemen als de heuphoek. Bekijk de instructievideo om te zien hoe dit bereikt kan worden.

Armlengte
Een belangrijk technisch aspect van de uitvoering van de zweefkip is het gestrekt houden van de armen gedurende de gehele oefening. Aangezien wij geen elleboog in het model hebben opgenomen, lijkt het op het eerste gezicht niet mogelijk om daar zicht op te krijgen. We kunnen echter wel een maat bepalen die direct wordt beïnvloed door flexie in de elleboog: de armlengte. Maak, met behulp van de hierboven geleerde technieken een grafiek van de armlengte (de benodigde data staat al in kolom U). Zet deze grafiek ook weer op het 2^{e} tabblad. Bekijk vervolgens in kinovea of u de afname van de armlengte in de film en de grafiek kunt terugvinden.

De slingerende turner
Een van de opvallende zaken aan de kipbeweging is dat de turner er in slaagt om zijn lichaamszwaartepunt aan het eind van de beweging in een hogere positie (tov van de vloer) te brengen dan bij aanvang van de beweging. De hoogte van het

zwaartepunt vertegenwoordigt een bepaalde hoeveelheid energie (*potentiële gravitatie-energie*) Deze energie-inhoud wordt berekend met de **formule massa van de turner** (kg) maal **zwaartekrachtversnelling** (m/sec^2) maal **hoogte** (m) (m*g*h). M en g zijn uiteraard constant, de toename van de energie-inhoud is het gevolg van een toename van h. Deze extra energie krijgt de turner niet cadeau. Die heeft hij er onderweg zelf in moeten stoppen. Als de turner zich had beperkt tot simpelweg passief heen en weer slingeren was hij nooit in deze eindpositie beland maar onder de rekstok geëindigd. Om te snappen wat de turner precies doet kijken we eerst naar de wetmatigheden van de energie-uitwisseling bij een slingerbeweging

Energie-uitwisseling
Indien een slingerende massa aan een touwtje geen enkele uitwendige weerstand zou ondervinden (geen luchtwrijving of frictie in het draaipunt) zou het eeuwig blijven slingeren en steeds naar de aanvangshoogte terugkeren. Blijkbaar wordt de energie behouden. De slinger verliest vanaf het hoogste punt welliswaar potentiële gravitatie-energie maar krijgt daar een andere vorm van energie voor terug (snelheidsenergie of kinetische energie). Zodra de slinger door het laagste punt heen draait vindt het omgekeerde plaatst. Kinetische energie wordt omgezet in potentiële energie.

Hieronder is een animatie te zien van zo'n ideale wrijvingsloze slinger (ctrl+click op de afbeelding om de link te openen). Bestudeer de uitwisseling van energie die daar plaatsvindt. De totale energie-inhoud van het systeem (rechter balk) blijft constant. De bijdrage van de beide energievormen aan dit totaal is echter aan veranderingen onderhevig.



Lichaamszwaartepunt
Een turner kan niet zonder verliezen slingeren, maar kan (in tegenstelling tot de passieve slinger uit het vorige voorbeeld) wel energie toevoegen (door

spiercontracties). We gaan de wijze waarop de turner dit doet onderzoeken door de beweging van de turner op te vatten als een slingerende massa (het lichaamszwaartepunt van de turner) met een variabele slingerlengte (de afstand van het zwaartepunt tot de rekstok (hand). We analyseren, met behulp van dit model, vervolgens de uitwisseling tussen potentiële en kinetische energie van de turner. De eerste stap in dit modelleerproces is dan ook logischerwijs het bepalen van de positie van het lichaamszwaartepunt in elk frame van de video. Om die positie te vinden moeten we per frame weten waar alle deelmassa's zich bevinden en hoe groot deze deelmassa's zijn. Met behulp van die informatie kunnen we dan vervolgens in elk frame de positie van het totale zwaartepunt berekenen. Nu kunnen we de lichaamsdelen van de turner natuurlijk niet afzonderlijk wegen. We moeten ons voor een schatting van de massaverdeling daarom laten leiden door de literatuur (zie tabel van Dempster).
http://www.health.uottawa.ca/biomech/csb/Archives/dempster.pdf

In de tabel van dempster staat aangegeven hoe het gewicht van elk lichaamsdeel zich verhoud tot het totale lichaamsgewicht (in een fractie van 1 uitgedrukt) en waar het zwaartepunt van dit lichaamsdeel zich bevindt (uitgedrukt in een fractie van de totale segmentlengte). Met behulp van die info maken we in kolom X een lijstje zoals hier onder: In kolom y zetten we de bijbehorende waarden.



In Y1: 75 in Y2: =Y1*0,1 in Y3: = Y1*0,58 en in cel Y4: =Y1*0,32 Met behulp van deze code wordt in elke cel de deelmassa berekend op basis van het lichaamsgewicht. Let op de waarde in cel Y2 vertegenwoordigt het gewicht van twee armen en cel Y4 het gewicht van twee benen. De posities van de deelzwaartepunten komen in de cellen Y5 t/m 7. Deze getallen staan voor de verhouding tussen de lengte van het element en afstand van het zwaartepunt (vanaf proximaal). Voor de romp hebben we het zwaartepunt wat verder naar proximaal gelegd dan de data van Dempster aangeeft, aangezien het hoofd wel in de totale massa is opgenomen maar niet als element in ons model voorkomt. Noteer in cel Y5: 0,45 in Y6: 0,4 en in Y7: 0,45.
We gaan de cellen van kolom Y die we zojuist gevuld hebben een eigen naam geven. Selecteer cel Y1 en Type in het naamvak van deze cel (zie onderstaande afbeelding) BW (gevolgd door een enter) Deze cel heeft nu voortaan BW (staat voor Body Weight) als naam. Het voordeel daarvan zal blijken zodra we gaan rekenen met de massa's en de massaposities.



Geef cel Y2 tm Y7 ook een eigen naam. Hanteer daarbij de volgende namen (exact overnemen incl hoofdletters)

Marm (cel Y2)
Mromp (cel Y3)
Mbeen (cel Y4)
Posarm (cel Y5)
Posromp (celY6)
Posbeen (cel Y7)

Met behulp van bovenstaande informatie kunnen we gaan rekenen. We hebben de zwaartepunten van de proefpersoon niet rechtstreeks gemeten in de video, maar om er berekeningen mee te kunnen doen moeten we, net als van de markers, de x en y coördinaat van alle deelzwaartepunten in elk frame van de video wel kennen. We gaan deze x en y coördinaten daarom bereken en toevoegen aan onze gemeten data.



Zwaartepunt armen

In de figuur hiernaast zien we de turner in frame 1 van de video. De witte stippen zijn de deelzwaartepunten die we willen berekenen. We kennen de coördinaten van alle rode punten waaronder de schouder en hand (afgerond weergegeven) en de schouder en heuphoeken. We kennen ook de armlengte want die hebben we berekend in kolom U. Toch is dit onvoldoende informatie om de positie van het deelzwaartepunt van de armen te berekenen. We kunnen wel de lengte van het lijnstuk schouder naar zwaartepunt armen berekenen in kolom Z.

Noteer in cel Z3 =U3*Posarm. Gebruik de muis-sleeptruc om deze formule ook op alle overige cellen van kolom Z toe te passen. Klik hierna eens op cel Z10 en bestudeer de formule. Zoals U ziet wordt de formule van cel U3 hier toegepast maar is Excel zelf zo slim om te begrijpen dat u in cel Z10 de waarde uit U10 wilt gebruiken in de berekening en Niet de waarde van cel U3 die u in cel Z3 had opgegeven. De formule gebruikt echter wel steeds de cel Posarm. Door een naam te geven aan een cel in Excel hanteert Excel een zogenaamde vaste celverwijzing indien de betreffende cel wordt opgenomen in een berekening. Hadden we de cel geen naam gegeven, dan had Excel in cel Z10 de volgende berekening uitgevoerd =U10*Y10. Vaste celverwijzingen kunnen ook wel worden gemaakt zonder de cel een naam te geven. Indien u in bv cel C4 een vaste celverwijzing wilt maken dan moet in een formule met C4 naar de cel verwezen worden. De dollar ($) tekens maken dat Excel een vaste celverwijzing toe zal passen. Het geven van een naam aan een cel heeft echter als voordeel dat de formule wat beter interpreteerbaar wordt.

We noemen de vector van de schouder naar het zwaartepunt van de armen z. De lengte van deze vector kennen we nu. We zijn op zoek naar de waarde van zx en zy.

Als we die kunnen bepalen en die waarden vervolgens optellen bij de x en y coördinaten van de schouder, hebben we de positie van het zwaartepunt vastgelegd.



Bij een bepaalde lengte van z kunnen zx en zy variëren. Deze variatie wordt bepaald door de hoek die de vector maakt met de horizontaal. Aangezien de vector z altijd op de verbindingslijn schouder-hand zal liggen is de hoekstand van de vector z tov de horizontaal gelijk aan die van de arm met de horizontaal. Die hoek (arm tov horizotaal) kunnen we met behulp van een hulpvector en het inproduct berekenen.



We geven de horizontale hulpvector d de lengte 1 De coördinaten van het eindpunt van deze hulpvector zijn dus: (90,158). Vervolgens vullen we de waarden van de hulpvector (dx en dy en d) in, in de formule van het inproduct (die we eerder hebben gebruikt om de heup en schouderhoek te berekenen):

$$\alpha = \text{acos}\ \frac{(cx.dx)+(cy.dy)}{c.d}$$

Waarden van de hulpvector ingevuld:

$$\alpha = \operatorname{acos} \frac{(cx.1)+(cy.0)}{c.1}$$

$$\alpha = \operatorname{acos} \frac{cx}{c}$$

De formule van het inproduct wordt door het gebruik van een horizontale hulpvector met lengte 1 vereenvoudigd tot de bekende relatie in een rechthoekige driehoek: cos α = aanliggende/hypotenusa

De waarden van cx en c (armlengte) hebben we al eerder berekend in kolom S en U. We kunnen dus de hoek van de arm met de horizontaal bereken in cel AA3

=GRADEN(BOOGCOS(S3/U3)) of (engels) =DEGREES(ACOS(S3/U3))

Merk op dat we in deze formule de omrekening van radialen naar graden direct in één formule combineren. In kolom V en W deden we dit nog afzonderlijk. Pas de formule weer op de gehele kolom toe en maak een grafiek van het resultaat op het 1^{e} tabblad.



De grafiek van de hoek van de arm met de horizontaal vertoont een merkwaardige omslag rond frame 76. Bekijk in kinovea de video en bestudeer wat er zich daar voordoet.
De omslag vindt plaats zodra de schoudermarker boven de handmarker uitkomt. De rekenregel in kolom AA3 blijft gewoon de hoek tussen de hulpvector en de arm berekenen en die loopt, na de passage door 0, natuurlijk weer op. Voor ons doel is dat echter niet handig. Neem bijvoorbeeld de hoekwaarde 40. Deze komt driemaal voor in de bovenstaande grafiek. De eerste twee keer dat deze hoekwaarde voorkomt bevindt de arm zich inderdaad in dezelfde positie Maar de derde maal bevindt de arm zich veertig graden boven de horizontaal en niet eronder. De hoeken zoals ze nu in de grafiek voorkomen zijn dus geen bruikbare maat om de positie van

het zwaartepunt te bepalen. Zodra de schouder boven de hand komt zou de hoekwaarde negatief moeten worden weergegeven. Om dit te bereiken kunnen we natuurlijk gewoon een – teken in alle cellen zetten voorbij het punt waarop dit gebeurt, maar dat maakt ons rekenwerk niet langer bruikbaar voor een ander filmpje waarin het natuurlijk niet zeker is dat deze gebeurtenis in hetzelfde frame plaatsvindt. We moeten de detectie van deze gebeurtenis dus automatiseren.
Excel beschikt daarvoor over een aantal zogenaamde voorwaardelijke functies. Wij maken gebruik van de als/dan functie (engels if/then)

De voorwaarde waarop we ons besluit over het – teken voor de hoek willen baseren is de relatie tussen de hand en de schouder y coördinaat. In woorden komt de overweging neer op: Als de y waarde van de schouder hoger is dan die van de hand zet dan een – voor de hoek, toon in alle andere gevallen gewoon de hoek. In excelcode ziet deze afweging er als volgt uit

=ALS(F3>H3;-AA3;AA3)



Bij het invoeren van dergelijke complexere formules kun je gebruik maken van de formule-wizzard van Excel. Selecteer cel AB3. Klik op het fx symbool en type de gewenste functie in het invoerveld of kies deze uit de lijst (als of if)

In het scherm dat nu verschijnt kun je de logische test F3>H3



in het bovenste veld invullen. In het veld **waarde als onwaar** verwijs je naar de cel AA3.

In het veld **waarde als waar** type je –AA3 (dat is geen cel waar je naar kunt verwijzen, maar een bewerking van een cel). Sluit af met ok. Pas de formule toe op de gehele kolom

Maak een grafiek en bekijk het verschil met de voorgaande grafiek (die je daarna kunt verwijderen)



Nu we de hoek van de arm met de horizontaal hebben bepaald, kunnen we de coördinaten van het zwaartepunt van de arm berekenen. Door de vector z (naar het zwaartepunt van de armen) en de hulpvector d te gebruiken (met hun inmiddels bekende hoek) In onderstaande formules zijn α dx,dy, d en z bekend. Op basis daarvan kunnen we zx en zy uitrekenen.

$$\cos\alpha = \frac{(zx.dx)+(zy.dy)}{z.d}$$

$$\cos\alpha = \frac{(zx.1)+(zy.0)}{z.1}$$

$$\cos\alpha = \frac{zx}{z}$$

$zx = z*\cos\alpha$
$zy = z*\sin\alpha$

De x waarde van het zwaartepunt berekenen we in cel AC3:

=E3+Z3*COS(RADIALEN(AB3)) of (engelse versie Excel)
=E3+Z3*COS(RADIANS(AB3))

E3 is de waarde van de x coördinaat van de schouder. De term radialen in de formule rekent de hoekwaarden in graden van cel AB3 om naar radialen. excel verwacht namelijk hoekwaarden in radialen om de cosinus te bepalen.

Bepaal zelf de formule voor de y coördinaat en zet deze in AD3. Raadpleeg zonodig de appendix.

Selecteer de data in beide kolommen en kies in het menu invoegen (insert)



voor het grafiektype spreiding (scatterplot) en kies vervolgens de eerste optie op de tweede rij (vloeiende lijnen). De grafiek die nu verschijnt toont een x-y plot van het zwaartepunt van de armen. De slingerende beweging rond de rekstok is duidelijk herkenbaar.



Zwaartepunt romp
De lengte van de vector van de schouder naar het zwaartepunt van de romp is te berekenen als =M3*Posromp Noteer deze formule in cel AE3 en pas hem toe op alle datapunten.



hoek β tussen de hulpvector(d) en de rompvector (a) is in cel AF3 te berekenen met:

=GRADEN(BOOGCOS(Q3/M3))
Gebruik DEGREES en ACOS voor de engelse versie van excel

(vul zelf de bekende waarden weer in in de inproduct formule om te doorzien hoe bovenvermelde formule is afgeleid.)

De coördinaten van het zwaartepunt van de romp vinden we met:

ZrompX =E3+AE3*COS(RADIALEN(AF3)) (in kolom AG)

ZrompY=F3-AE3*SIN(RADIALEN(AF3))   (in kolom AH)

De grafiek van de hoek van de romp met de horizontaal ziet er als volgt uit: Omdat de heupmarker nergens in de video boven de schoudermarker uitkomt hoeven we in dit geval gen als (if) functie aan deze kolom toe te voegen. Maar als u later nog eens een andere turner zou filmen waarbij dit wel gebeurt, zou het wel nodig kunnen zijn.



De xy plot van de coördinaten van het zwaartepunt van de romp:



Zwaartepunt benen

De lengte van de vector van de heup naar het zwaartepunt van de benen is te berekenen als =N3*Posbeen

Noteer deze formule in cel Ai3 en pas hem toe op alle datapunten.

In AJ3 de hoek van het been met horizontaal

=GRADEN(BOOGCOS(K3/N3))

Net als bij de armen speelt hier ook weer het probleem van de hoekdefinitie op het moment dat de y waarde van de voet die van de heup overstijgt. Controleer dit door de grafiek te bestuderen in combinatie met de bijbehorende videoframes in kinovea.

Bedenk zelf in kolom AK de oplossing met behulp van een als / if functie in Excel. (Raadpleeg zo nodig de appendix van deze handleiding)



De juiste grafiek van de hoek staat ter controle hiernaast.

De coördinaten van het zwaartepunt van de benen vinden we met:

ZbeenX =C3+AI3*COS(RADIALEN(AK3)) (in kolom AL)

ZbeenY= Bedenk zelf de formule (in kolom AM) (raadpleeg zo nodig de appendix)



Ter controle hiernaast de grafiek (x-y plot)

Positie van het totale lichaamszwaartepunt.

Nu alle deelzwaartepunten bekend zijn, berekenen we de positie van het lichaamszwaartepunt. Ook hierbij moeten we weer zowel de x als de y coördinaat bepalen.

$$X_{lzp} = \frac{\sum_{i=1}^{n} \left(M_{seg\,i} * X_{seg\,i}\right)}{M_{body}}$$

$$Y_{lzp} = \frac{\sum_{i=1}^{n} \left(M_{seg\,i} * Y_{seg\,i}\right)}{M_{body}}$$

De formules waarmee we de x en y coördinaat van het lzp kunnen bepalen staan hiernaast. Vertaald komt deze formule neer op het volgende: De x coördinaat van het zwaartepunt vinden we door van alle segmenten (1 tm n) de massa van het betreffende segment te nemen en die te vermenigvuldigen met de x coördinaat van het zwaartepunt van dat segment. De som van al die producten (aangegeven met het sigma Σ symbool) delen we vervolgens door het totale lichaamsgewicht. Op vergelijkbare wijze vinden we de y coördinaat.

Een voorbeeld

In de figuur hieronder worden de afgeronde coördinaten van de zwaartepunten in de beginpositie getoond. Indien alle zwaartepunten in gelijke mate mee zouden wegen zouden we de x coördinaat van het LZP eenvoudig kunnen bepalen door alle x coördinaten op te tellen en de uitkomst door 3 te delen. In werkelijkheid is de bijdrage van elk zwaartepunt aan het geheel natuurlijk niet gelijk. We moeten een weegfactor gebruiken. Daartoe vermenigvuldigen we de x coördinaat van elke deelmassa met zijn 'gewicht'. Vervolgens tellen we die drie uitkomsten bij elkaar op en delen die door het lichaamsgewicht.



In dit frame:

(67*24)+(78*43,5)+(118*7,5)/75 = 78
(x coördinaat lzp in frame 1)

Voor de y coördinaten doen we hetzelfde

(75*24)+(141*43,5)+(164*7,5)/75 = 122
(y coordinaat lzp in frame 1)

De formule in excel:

Voor de x coördinaat =((AC3*Marm)+(AG3*Mromp)+(AL3*Mbeen))/BW
(in kolom AN)
Merk op dat hier vaste celverwijzingen worden toegepast.

Bedenk zelf de formule voor de y coördinaat (raadpleeg zo nodig de appendix).

Hieronder de xy plot van de baan van het lichaamszwaartepunt



Potentiële (gravitatie) energie van het zwaartepunt.

Een massa bezit ten opzichte van een grondvlak energie indien de massa zich in een zwaartekrachtveld boven dit vlak bevindt. Dat dit zo is merkt u als u bv van een stoel afspringt. Deze vorm van energie is te berekenen met de formule m*g*h (m=massa in kg, g = zwaartekrachtversnelling (9,82m/s^2) en h = hoogte in meters. In de beginpositie bezit de turner dus al potentiële energie. Zijn zwaartepunt bezit al een zekere hoogte. De y coördinaat van het lichaamszwaartepunt in frame 1 staat in cel AO3. Maar deze hoogte is natuurlijk bepaald ten opzichte van de linker-onderhoek van de video (daar hebben we de oorsprong van het assenstelsel gelegd). Het is echter realistischer om de hoogte ten opzicht van de vloer als maat te nemen. We gebruiken daarvoor de y coordinaat van de enkelmarker in de startpositie (te vinden in cel B3).
Noteer in cel X8: yvloer en in de cel daarnaast (cel Y8) verwijzen we naar de waarde in cel B3 =B3. Geef deze cel (de cel Y8 dus) de naam yvloer (om een vaste celverwijzing naar deze cel mogelijk te maken)

De formule voor de potentiële energie (in cel AP3) wordt dan:

=BW*9,82*((AO3-yvloer)/100). Vergelijk deze formule met de hierboven besproken berekening van Epot. Waarom wordt er door 100 gedeeld?



Maak een grafiek van het Epot (op de y as) tegen de framenummers (op de x as). Het resultaat wordt hiernaast getoond.

Bekijk de grafiek en bestudeer gelijktijdig de film in kinovea

De turner bezit in stand al potentiele energie (zijn zwaartepunt bevindt zich hoger dan het 0 niveau dat we hebben bepaald). Hij verliest en her wint aanvankelijk wat Epot door het inveren en opspringen voor aanvang van de zwaai. Zodra de zwaai is ingezet is er enige tijd sprake van verlies van Epot. Ga na met welk frame in de film het laagste punt in de Epot grafiek correspondeert. Vanaf frame 65 is er een snelle toename van Epot tot ruim boven het aanvangsniveau. Dat kan natuurlijk nooit passief slingeren zijn. In het meest ideale geval (indien er geen wrijving of andere verliezen zouden bestaan) zou de turner bij het passief terugzwaaien zijn zwaartepunt weer op de aanvangshoogte kunnen terugbrengen. Hij bereikt echter een hogere positie. De turner weet dus energie toe te voegen. Bestudeer de grafiek in combinatie met de film en de dataset en ga na welk datapunt correspondeert met de tweede maal dat het zwaartepunt van de turner zich loodrecht onder de stang (en dus de hand) bevindt (dus tijdens de terugzwaai). Vergelijk de hoogte van het zwaartepunt bij de eerste passage met de tweede. Dit verschil is ook goed te zien aan de x-y plot van het zwaartepunt.
Voor een compleet beeld moeten we ook de andere energievorm (kinetische energie) in de analyse betrekken.

Kinetische energie
De kinetische energie van de turner wordt berekend met: ekin = 0,5 * M * V^2
Hierin staat M voor de massa (kg) en V voor de snelheid (m/sec). de snelheid van de turner (zijn zwaartepunt) kennen we op dit moment nog niet. We hebben slechts coördinaten (posities). Snelheid is te berekenen als de verandering van de positie (delta x of y genoteerd als: Δx of Δy) gedeeld door de tijd waarin die positieverandering heeft plaatsgevonden; delta t (Δt). De verandering van positie in x en y richting vinden we door opvolgende coördinaten van elkaar af te trekken. De tijd halen we uit de framerate van de video. Bij 30 fps frames per seconde is de tijdstap tussen de beeldjes 1/30.

Deze methode wordt numeriek differentiëren genoemd. Om redenen die buiten het kader van deze handleiding vallen, is het gebruikelijk om de Δx of Δy niet over één maar over twee frames te bepalen en dan natuurlijk ook te delen door tweemaal Δt. Zie het voorbeeld in de tekening. De zo berekende waarde van de snelheid wordt vervolgens toegekend aan de middelste waarde (x).



In cel X9 zeten we de tekst framerate, in Y9 de bijbehorende waarde (30) In cel X10 de tekst dt en in cel Y10 berekenen we de bijbehorende waarde =1/Y9. Geef de cel Y10 de volgende naam: dt

Indien je in de toekomst een ander filmpje van een zweefkip wilt analyseren met een andere framerate volstaat het om de waarde in cel Y9 aan te passen.

In cel AQ4 (**let op niet in cel 3**) noteren we de excelformule: =((AN5-AN3)/100)/(2*dt) Probeer de berekeningswijze, zoals voorgeteld in de afbeelding, in deze formule te herkennen. Begrijpt u waarom er door 100 wordt gedeeld? In cel AQ 3 kunnen we deze formule niet toepassen omdat er geen voorafgaande x waarde (in cel 2) beschikbaar is. Bij het berekenen van een afgeleide treedt dit probleem altijd op. Ook aan het eind van de datareeks moeten we om dezelfde reden een cel eerder stoppen met de berekening.

Bedenk zelf de formule voor de snelheid in y richting in cel AR4. (raadpleeg zo nodig de appendix)

De werkelijke snelheid
We hebben de snelheid in x en y richting berekend en moeten deze twee componenten nu combineren tot de werkelijke snelheid. Bedenk zelf hoe dit gedaan moet worden, ontwerp de formule en pas die toe (vul daarmee kolom AS).
(Raadpleeg zo nodig de appendix)

De kinetische energie kan nu berekend worden. Bedenk zelf de excelformule en pas deze toe in kolom AT. Geheugensteuntje: ekin = 0,5 * M * V^2
(Raadpleeg zo nodig de appendix)

De totale energie-inhoud van het zwaartepunt.
Bij de animatie van de slinger zagen we dat de totale energie-inhoud van het

systeem constant bleef, er was slechts sprake van omzetting van energievormen. Denk je dat dit in het geval van de turner ook zo is? Test je voorspelling door de totale energie-inhoud te bereken (de optelsom van Epot en Ekin). Bereken dit in kolom AU (vanaf cel 3) en maak een grafiek waarin zowel Epot, Ekin en Etot getoond worden.



Analyse

De totale energie-inhoud van de turner blijkt vanaf de start van de zwaai (frame 21) eerst af te nemen om vervolgens toe te nemen tot boven het oorspronkelijke niveau. De turner eindigt immers in een hogere positie dan hij is gestart. Zijn potentiële energie bedroeg bij aanvang 703 Nm en in de eindpositie 1055 Nm. *(door het tracken en filteren van data kunnen er kleine verschillen bestaan tussen uw data en de voorbeelddata in deze handleiding)* Een toename van 352 Nm. Met behulp van de formule van de potentiële energie kunnen we dit verschil uitdrukken in hoogtewinst van het zwaartepunt. Epot = M*g*h omgeschreven levert dit op h = Epot/M*g. Invullen van waarden: h= 352/75*9,82 h (de hoogtetoename van het zwaartepunt) is 47,7 cm. Controleer of dit verschil in overeenstemming is met het verschil in de ywaarde van het zwaartepunt (eindwaarde min startwaarde).

In het traject tussen frame 21 en 35 weet de turner het energieverlies aardig te beperken (de potentiële energie wordt vrijwel geheel omgezet in kinetische energie. In de fase die dan volgt treedt er echter een duidelijk verlies op (tot frame 50). Daarna weet de turner veel energie toe te voegen.

Hoe de turner er in slaagt om energie toe te voegen is een gecompliceerd verhaal. De situatie vertoont veel overeenkomsten met de wijze waarop we er in slagen om hoogte te winnen op een schommel in de speeltuin. Er zijn twee methoden die daarbij gebruikt kunnen worden. Indien we staan op de schommel moeten we het zwaartepunt naar het draaipunt van de schommel toe brengen in de fase waarin de schommel door de verticale positie draait. Een zittende schommelaar maakt gebruik van een ander mechanisme.



staand schommelen



Zittend schommelen

Aan het eind van de voorwaartse zwaai wordt het lichaam voorovergekanteld, aan het eind van de achterzwaai juist achterover.
Voor meer info en instructieve filmpjes zie [deze internet link]

Beide methoden worden gebruikt door de turner. Hij draait snel voorover en brengt tevens zijn zwaartepunt dichter bij het draaipunt. Op deze wijze voegt hij energie toe en kan hij in een hogere positie eindigen dan waarin hij begon.

Om een beeld te krijgen van de inspanning die de turner daarbij levert, bepalen we het vermogen dat wordt geleverd tijdens de oefening. Het vermogen dat de turner levert is de verandering van de totale energieinhoud per tijdseenheid. Een voorbeeld:

U tilt een last van 20 kg een meter omhoog in 1 seconde. We stellen de zwaartekrachtversnelling voor het gemak even op 10 m/sec^2 . De arbeid die u levert is gelijk aan de toename van de energie van de last (m.g.h) = 200 Nm. Het vermogen is arbeid gedeeld door tijd = 200 watt (Nm/sec). Als u dezelfde klus in een halve seconde zou uitvoeren is het vermogen dat u moet leveren echter dubbel zo groot; 400 watt. Indien u de last zou laten zakken over een meter levert u negatief vermogen. De snelheid waarmee de energie-inhoud van het zwaartepunt wijzigt is dus een goed maat voor de inspanning van de turner.

We kunnen natuurlijk het gemiddelde vermogen dat hij leverde gedurende de oefening berekenen door het verschil in energie-inhoud (eind min start) te delen door de tijd van de gehele oefening. = 352/3,4 = 103 watt. Deze berekening is echter een schromelijke onderschatting van het werkelijke vermogen. De turner verliest immers eerst energie en moet dus in een korte tijd veel meer energie toevoegen dan bovenstaande berekening doet vermoeden. We moeten dus eigenlijk niet het gemiddelde vermogen berekenen, maar het vermogen over elke tijdstap bepalen.

Deze verandering van de energie-inhoud (Etot) in de tijd tussen twee frames (het momentane vermogen dat de turner levert) kunnen we berekenen dmv numeriek differentiëren. Die methode hebben we bij het berekenen van de snelheid van het zwaartepunt (in x en y richting) al eens gebruikt.

De formule voor het vermogen in cel AV4:
=(AU5-AU3)/(2*dt)
Plot de data weer in een grafiek.



De werkelijke vermogens liggen aanzienlijk hoger dan het gemiddelde vermogen suggereerde. Bestudeer de grafiek van het vermogen en die van de totale energie-inhoud. Daar waar de totale energie-inhoud snel stijgt worden de grootste vermogens geleverd. Grafieken als deze (verkregen door numeriek differentiëren) zijn zeer gevoelig voor filteren. Indien u de data meer of minder filtert heeft dit aanzienlijke invloed op de uitkomsten. Om vergelijkingen tussen verschillende uitvoeringen mogelijk te maken moet de filtermethode dus ongewijzigd blijven.

Efficiëntie
In deze handleiding heeft u een groot aantal data-bewerkingen geleerd die ook op andere bewegingsvormen kunnen worden toegepast. Er ontbreekt er eigenlijk nog een: numeriek integreren. Met deze techniek wordt het oppervlak onder een grafiek bepaald.



$$Area = (t_2 - t_1)\left[\frac{f(t_1) + f(t_2)}{2}\right]$$

Het oppervlak onder een curve kan worden benaderd door de zogenaamde trapeziumregel toe te passen. Het oppervlak onder de curve tussen twee opvolgende tijdstippen (t_1 en t_2) wordt daarbij opgevat als het oppervlak van de trapeziumvormige figuur onder de curve. (in de linker figuur het gele plus het paarse oppervlak). Natuurlijk is dit een benadering de grafiek loopt tussen de twee punten niet echt recht. Maar bij heel kleine tijdstapjes is deze benadering nauwkeurig genoeg. Dit trapeziumvormige oppervlak komt (qua totaal oppervlak) overeen met de optelsom van het gele en het groene oppervlak van de rechter figuur. Dit oppervlak is te berekenen als het product dat wordt weergeven in de formule (Area).Hierbij moet dus steeds het oppervlak van de voorgaande tijdstap worden opgeteld. Het eindresultaat is dan ook geen serie van uitkomsten, maar één enkel getal dat gelijk is aan het oppervak onder de curve.

We zullen deze techniek gebruiken om de negatieve en de positieve vermogensleverantie van de turner met elkaar te vergelijken. We bepalen eerst alle deelopervlakken onder de gehele vermogenscurve (positief en negatief).

In cel AW4 en daarna toepassen op de hele kolom

=((AV5+AV4)/2)*dt

In cel AX3 berekenen we de optelsom van alle positieve oppervlakken uit kolom AW

=SOM.ALS(AW4:AW103;">0") (sumif in de engelse versie)

In de cel daaronder de optelsom van alle ‘negatieve’ oppervlakdelen

=SOM.ALS(AW4:AW103;"<0") (sumif in de engelse versie)

Tenslotte tellen we in de cel daaronder beide getallen bij elkaar op. Deze optelsom levert ons het oppervlak van de netto positie vermogens.

Merk op dat de uitkomst van deze laatste optelsom gelijk is aan de eerder berekende toename van energie van de turner. Dit is logisch want het oppervlak onder de netto (positieve) vermogenscurve is het product van vermogen en tijd (Nm/sec * sec) = Nm = arbeid en dat is natuurlijk weer gelijk aan de toename van de energie-inhoud van de turner.
om de verhouding tussen positieve en negatieve arbeidsleverantie te berekenen bepalen we in cel AX6 de verhouding tussen de totaal geleverde arbeid en de netto positieve arbeid (in procenten)

=(AX5/AX3)*100

Het rendement is in dit geval dus ongeveer 50%. Ongeveer de helft van de geleverde arbeid is ten goede gekomen aan de hoogtewinst.

Bedenk zelf hoe deze analyse ook uitgevoerd zou kunnen worden met behulp van de informatie uit de curve van de totale energie-inhoud. (vergelijk het startniveau met het laagste en het hoogste (eind) niveau).

Tot zo ver deze handleiding. Veel succes met uw eigen analyses.

Aad Lagerberg

www.damcursus.nl

Appendix

Formule voor Cx (in kolom S)
=G3-E3

Formule voor Cy (in kolom T)
=H3-F3

Formule voor de y coordinaat van het zwaartepunt arm (in kolom AD)
=F3+Z3*SIN(RADIALEN(AB3))

Formule hoek been met horizon (in Kolom AK)
=ALS(B3>D3;-AJ3;AJ3)

Y waarde van het zwaartepunt van het been(in kolom AM)
=D3-AI3*SIN(RADIALEN(AK3))

Y coördinaat van het lichaamszwaartepunt (in kolom AO):
=((AD3*Marm)+(AH3*Mromp)+(AM3*Mbeen))/BW

Snelheid zwaartepunt in y richting (in kolom AR)
=((AO5-AO3)/100)/(2*dt)

Snelheid van het lichaamszwaartepunt (in kolom AS):
=WORTEL(AQ4^2+AR4^2)

Formule voor de kinetische energie (in kolom AT)
=0,5*BW*(AS4)^2